



INSTITUTO FEDERAL DE
EDUCAÇÃO, CIÊNCIA E TECNOLOGIA
RIO GRANDE DO NORTE

AULA 10

COMANDOS DE REPETIÇÃO ANINHADOS

Disciplina: Algoritmos
Professora: Alba Lopes

alba.lopes@ifrn.edu.br

<http://docente.ifrn.edu.br/albalopes>

COMANDOS DE REPETIÇÃO ANINHADOS

- Em alguns casos, é preciso utilizar comandos de repetição aninhados/encadeados na resolução de problemas
 - Um comando de repetição dentro de outro
- É possível aninhar quantos laços forem necessários



EXEMPLO ENQUANTO

```
algoritmo "lacos_aninhados"
var
    i, j: inteiro
inicio
    i <- 1
    enquanto ( i <= 10 ) faca
        escreval("**Laço externo: ", i)
        j <- 1
        enquanto (j <= 10) faca
            escreval("Laço interno", j)
            j <- j + 1
        fimenquanto
        i <- i + 1
    fimenquanto
fimalgoritmo
```



EXEMPLO REPITA

```
algoritmo "lacos_aninhados"
var
    i, j: inteiro
inicio
    i <- 1
    repita
        escreval("**Laço externo: ", i)
        j <- 1
        repita
            escreval("Laço interno", j)
            j <- j + 1
        ate ( j > 10 )
        i <- i + 1
    ate ( i > 10 )
finalgoritmo
```



EXEMPLO PARA

```
algoritmo "lacos_aninhados"  
var  
    i, j: inteiro  
inicio  
    i <- 1  
    para i de 1 ate 10 faca  
        escreval("**Laço externo: ", i)  
        para j de 1 ate 10 faca  
            escreval("Laço interno", j)  
        fimpara  
    fimpara  
fimalgoritmo
```



EXEMPLO

- Um professor possui 5 turmas, e cada turma possui 40 alunos. Construa um algoritmo que leia a nota dos alunos de cada uma das turmas e exiba a média das notas por turma.

```
algoritmo "turmas"  
var  
  i, j: inteiro  
  nota, soma, media: real  
inicio  
  para i de 1 ate 5 faca  
    escreval("Digite as notas da turma ", i, ": ")  
    soma <- 0  
    para j de 1 ate 40 faca  
      escreval("Digite a nota do aluno ", j, ": ")  
      leia(nota)  
      soma <- nota + soma  
    fimpara  
    media <- soma / 40  
    escreval("A média das notas da turma ", i, " é: ", media)  
  fimpara  
finalgoritmo
```



EXEMPLO

Crie um algoritmo que leia um número do usuário e desenhe o seguinte padrão na tela.

- **Ex:** para o número = 5



EXEMPLO

Crie um algoritmo que leia um número do usuário e desenhe o seguinte padrão na tela.

```
algoritmo "padrao"  
var  
    i, j, numero: inteiro  
inicio  
    escreva("Digite um número:")  
    leia (numero)  
    para i de 1 ate numero faca  
        para j de 1 ate numero faca  
            escreva("*")  
        fimpara  
        escreval("")  
    fimpara  
fimalgoritmo
```



EXERCÍCIOS

1. Crie um algoritmo que leia um número do usuário e desenhe o seguinte padrão na tela.
 - **Ex:** para o número = 5

*

**



EXERCÍCIOS

2. Crie um algoritmo que leia um número do usuário e desenhe o seguinte padrão na tela.

Ex: para o número = 5

**

*



EXERCÍCIOS

3. Crie um algoritmo que simule o funcionamento de um caixa de supermercado. O caixa fica aberto até o fim do expediente e pode processar a compra de vários clientes.
 - Cada cliente pode comprar vários itens. Ao ler cada item deve ser exibida uma mensagem para o operador do caixa perguntando se há mais itens a serem processados. Ao final, exiba quanto a compra custou ao cliente. E então solicite do operador do caixa a informação se deseja fechar o caixa.
 - Quando o caixa for fechado, exiba quanto de dinheiro aquele caixa apurou no dia.



EXERCÍCIOS

4. Crie um algoritmo utilizando laços de repetição aninhados que exiba na tela a tabuada de multiplicação dos números de 1 a 9

